@echo off
setlocal enabledelayedexpansion

REM Step1 - 清空并重新创建 ../target/usermanager 目录
if exist ..\target\usermanager rmdir /s /q ..\target\usermanager
mkdir ..\target\usermanager

REM Step2 - 拷贝 webapp 的内容到 ../target/usermanager
xcopy /E /I ..\src\main\webapp\* ..\target\usermanager\

REM Step3 - 设置依赖路径为 WEB-INF/lib 下的所有 jar 文件
set DEPENDENCIES=
for /r ..\src\main\webapp\WEB-INF\lib %%g in (*.jar) do set DEPENDENCIES=!DEPENDENCIES!;%%g

REM Step4 - 编译 Java 文件，输出到 ../target/classes
@REM 方法1 : 查找所有 Java 文件并输出到 sources.txt
dir /s /B ..\src\main\java\*.java > sources.txt
javac -encoding UTF-8 -cp !DEPENDENCIES! -d ..\target\classes\ @sources.txt
del sources.txt

@REM 方法2 : 递归查找所有 Java 文件并编译
@REM for /r ..\src\main\java %%g in (*.java) do javac -encoding UTF-8 -cp !DEPENDENCIES! -d ..\target\classes\ %%g

REM Step5 - 拷贝编译后的 .class 文件到 ../target/usermanager/WEB-INF/classes
mkdir ..\target\usermanager\WEB-INF\classes
xcopy /E /I ..\target\classes\* ..\target\usermanager\WEB-INF\classes\

REM Step6 - 创建 WAR 文件 (可选)
cd ..\target\usermanager
jar -cvf ..\usermanager.war *
cd ..\..

endlocal